*как-то тут все поменялось*
Задача, наверное, обычная, но что-то никак. Есть допустим две таблицы:
реализации
Док Дата Сумма
Р1 01.10 1000
Р2 03.10 2000
Р3 05.10 3000

и оплаты, структура та же
О1 02.10 1500
О2 04.10 1000
О3 10.10 3000

хочу получить запрос, который бы выдал реализации и даты и суммы их оплат. Структура: док реализации, док оплаты, дата оплаты, сумма оплаты
Р1 О1 02.10 1000
Р2 О1 02.10 500
Р2 О2 04.10 1000
Р2 О3 10.10 500
Р3 О3 10.10 2500

реализации и оплаты я умею выстроить как угодно, с нарастающим итогом и без. Но как их дальше совместить? Могу перебором расписать, но запросом было бы технологичнее.
Нужно для отчета об оплаченных реализациях за период. База УТ 10.3.
---
Вам ссылку на статью дали, не понимаю, в чём трудность?:

```1c
    ВЫБРАТЬ
    "Р1" КАК Ячейка,
    1000 КАК Количество
    ПОМЕСТИТЬ Ячейки

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "Р2",
    2000

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "Р3",
    3000
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    "О1" КАК Товар,
    1500 КАК Количество
    ПОМЕСТИТЬ Товары

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "О2",
    1000

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "О3",
    3000
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Ячейки.Ячейка КАК Ячейка,
    Ячейки.Количество КАК Количество,
    ЕСТЬNULL(СУММА(Ячейки1.Количество), 0) + 1 КАК ПорядокРаспределенияС,
    ЕСТЬNULL(СУММА(Ячейки1.Количество), 0) + Ячейки.Количество КАК ПорядокРаспределенияПо
    ПОМЕСТИТЬ ЯчейкиПоПорядку
    ИЗ
    Ячейки КАК Ячейки
    ЛЕВОЕ СОЕДИНЕНИЕ Ячейки КАК Ячейки1
    ПО Ячейки.Ячейка > Ячейки1.Ячейка

    СГРУППИРОВАТЬ ПО
    Ячейки.Ячейка,
    Ячейки.Количество
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Товары.Товар КАК Товар,
    Товары.Количество,
    ЕСТЬNULL(СУММА(Товары1.Количество), 0) + 1 КАК ПорядокРаспределенияС,
    ЕСТЬNULL(СУММА(Товары1.Количество), 0) + Товары.Количество КАК ПорядокРаспределенияПо
    ПОМЕСТИТЬ ТоварыПоПорядку
    ИЗ
    Товары КАК Товары
    ЛЕВОЕ СОЕДИНЕНИЕ Товары КАК Товары1
    ПО Товары.Товар > Товары1.Товар

    СГРУППИРОВАТЬ ПО
    Товары.Товар,
    Товары.Количество
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ЯчейкиПоПорядку.Ячейка КАК Ячейка,
    ТоварыПоПорядку.Товар КАК Товар,
    ВЫБОР
    КОГДА ТоварыПоПорядку.ПорядокРаспределенияПо < ЯчейкиПоПорядку.ПорядокРаспределенияПо
    ТОГДА ТоварыПоПорядку.ПорядокРаспределенияПо
    ИНАЧЕ ЯчейкиПоПорядку.ПорядокРаспределенияПо
    КОНЕЦ - ВЫБОР
    КОГДА ТоварыПоПорядку.ПорядокРаспределенияС > ЯчейкиПоПорядку.ПорядокРаспределенияС
    ТОГДА ТоварыПоПорядку.ПорядокРаспределенияС
    ИНАЧЕ ЯчейкиПоПорядку.ПорядокРаспределенияС
    КОНЕЦ + 1 КАК Количество
    ИЗ
    ЯчейкиПоПорядку КАК ЯчейкиПоПорядку
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ТоварыПоПорядку КАК ТоварыПоПорядку
    ПО ЯчейкиПоПорядку.ПорядокРаспределенияС <= ТоварыПоПорядку.ПорядокРаспределенияПо
    И ЯчейкиПоПорядку.ПорядокРаспределенияПо >= ТоварыПоПорядку.ПорядокРаспределенияС
```

Результат:
Ячейка Товар Количество
Р2 О2 1 000
Р1 О1 1 000
Р2 О1 500
Р2 О3 500
Р3 О3 2 500
---
как оплата связана с реализацией ?
---
в контексте задачи это неважно, есть две таблицы, в таблице оплат - то, что относится к таблице реализаций. Вообще-то они по заказам, но это неважно.

(3) см. выше, по заказам, но это неважно для решения задачи.
---
(1)
без четкого условия связи между реалами и оплатами, корректного результата не будет.
В оплате, должен быть реквизит сделка, вот через него и нужно связывать таблицы, левым соединением.
А так, будет вообще непонятно...с чем соединять, особенно если речь идет о выявлении неоплаченных реалов.
Т.е. дата может отличатся, да еще и сам факт наличия оплаты либо есть, либо нет.
В таблицах - информации для связи недостаточно. Это для выходных данных, для пользователя можно уменьшать количество колонок,
но алгоритм связывающий, эти две выборки, должен работать с бОльшим количеством данных.
---
условие связи отработано при заполнении таблиц. Естественно оно есть, но оно уже учтено до этой задачи.
Простейший пример: взаиморасчеты по заказам. По каждому заказу несколько реализаций и несколько оплат, суммы и даты и оплат и отгрузок распределяются произвольно, как получится.
Таблицы с отгрузками и оплатами мы заполнили, дальше остается их связать, чтобы получить суммы и даты оплаты каждой отгрузки.
Вот как связать, что-то не доходит до меня.
---
условие связи отработано при заполнении таблиц - и как это видно ?
---
это сказано в условии задачи...... ребят, если нет по делу, может, не будем забивать тему? действительно нужно решение задачи, которая в (1) описана полностью со всеми условиями.
---
по заказам ведете ? или по расчетным документам ?
---
Левым соединением совместить.
---
Недавно публикация была тут http://infostart.ru/public/568299/
Вроде как раз по теме
---
да, похоже, но что-то никак не приложится к моему примеру...
---
Вам ссылку на статью дали, не понимаю, в чём трудность?:

```1c
    ВЫБРАТЬ
    "Р1" КАК Ячейка,
    1000 КАК Количество
    ПОМЕСТИТЬ Ячейки

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "Р2",
    2000

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "Р3",
    3000
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    "О1" КАК Товар,
    1500 КАК Количество
    ПОМЕСТИТЬ Товары

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "О2",
    1000

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "О3",
    3000
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Ячейки.Ячейка КАК Ячейка,
    Ячейки.Количество КАК Количество,
    ЕСТЬNULL(СУММА(Ячейки1.Количество), 0) + 1 КАК ПорядокРаспределенияС,
    ЕСТЬNULL(СУММА(Ячейки1.Количество), 0) + Ячейки.Количество КАК ПорядокРаспределенияПо
    ПОМЕСТИТЬ ЯчейкиПоПорядку
    ИЗ
    Ячейки КАК Ячейки
    ЛЕВОЕ СОЕДИНЕНИЕ Ячейки КАК Ячейки1
    ПО Ячейки.Ячейка > Ячейки1.Ячейка

    СГРУППИРОВАТЬ ПО
    Ячейки.Ячейка,
    Ячейки.Количество
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Товары.Товар КАК Товар,
    Товары.Количество,
    ЕСТЬNULL(СУММА(Товары1.Количество), 0) + 1 КАК ПорядокРаспределенияС,
    ЕСТЬNULL(СУММА(Товары1.Количество), 0) + Товары.Количество КАК ПорядокРаспределенияПо
    ПОМЕСТИТЬ ТоварыПоПорядку
    ИЗ
    Товары КАК Товары
    ЛЕВОЕ СОЕДИНЕНИЕ Товары КАК Товары1
    ПО Товары.Товар > Товары1.Товар

    СГРУППИРОВАТЬ ПО
    Товары.Товар,
    Товары.Количество
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ЯчейкиПоПорядку.Ячейка КАК Ячейка,
    ТоварыПоПорядку.Товар КАК Товар,
    ВЫБОР
    КОГДА ТоварыПоПорядку.ПорядокРаспределенияПо < ЯчейкиПоПорядку.ПорядокРаспределенияПо
    ТОГДА ТоварыПоПорядку.ПорядокРаспределенияПо
    ИНАЧЕ ЯчейкиПоПорядку.ПорядокРаспределенияПо
    КОНЕЦ - ВЫБОР
    КОГДА ТоварыПоПорядку.ПорядокРаспределенияС > ЯчейкиПоПорядку.ПорядокРаспределенияС
    ТОГДА ТоварыПоПорядку.ПорядокРаспределенияС
    ИНАЧЕ ЯчейкиПоПорядку.ПорядокРаспределенияС
    КОНЕЦ + 1 КАК Количество
    ИЗ
    ЯчейкиПоПорядку КАК ЯчейкиПоПорядку
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ТоварыПоПорядку КАК ТоварыПоПорядку
    ПО ЯчейкиПоПорядку.ПорядокРаспределенияС <= ТоварыПоПорядку.ПорядокРаспределенияПо
    И ЯчейкиПоПорядку.ПорядокРаспределенияПо >= ТоварыПоПорядку.ПорядокРаспределенияС
```

Результат:
Ячейка Товар Количество
Р2 О2 1 000
Р1 О1 1 000
Р2 О1 500
Р2 О3 500
Р3 О3 2 500
---
С датами это будет вот так:

```1c
    ВЫБРАТЬ
    "Р1" КАК Ячейка,
    1000 КАК Количество,
    ДАТАВРЕМЯ(2016, 10, 1) КАК Дата
    ПОМЕСТИТЬ Реализация

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "Р2",
    2000,
    ДАТАВРЕМЯ(2016, 10, 3)

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "Р3",
    3000,
    ДАТАВРЕМЯ(2016, 10, 5)
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    "О1" КАК Товар,
    1500 КАК Количество,
    ДАТАВРЕМЯ(2016, 10, 2) КАК Дата
    ПОМЕСТИТЬ Оплата

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "О2",
    1000,
    ДАТАВРЕМЯ(2016, 10, 4)

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    "О3",
    3000,
    ДАТАВРЕМЯ(2016, 10, 10)
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Реализация.Ячейка КАК Ячейка,
    Реализация.Количество КАК Количество,
    ЕСТЬNULL(СУММА(Реализация1.Количество), 0) + 1 КАК ПорядокРаспределенияС,
    ЕСТЬNULL(СУММА(Реализация1.Количество), 0) + Реализация.Количество КАК ПорядокРаспределенияПо,
    Реализация.Дата
    ПОМЕСТИТЬ РеализацияПоПорядку
    ИЗ
    Реализация КАК Реализация
    ЛЕВОЕ СОЕДИНЕНИЕ Реализация КАК Реализация1
    ПО Реализация.Ячейка > Реализация1.Ячейка

    СГРУППИРОВАТЬ ПО
    Реализация.Ячейка,
    Реализация.Количество,
    Реализация.Дата
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Оплата.Товар КАК Товар,
    Оплата.Количество,
    ЕСТЬNULL(СУММА(Оплата1.Количество), 0) + 1 КАК ПорядокРаспределенияС,
    ЕСТЬNULL(СУММА(Оплата1.Количество), 0) + Оплата.Количество КАК ПорядокРаспределенияПо,
    Оплата.Дата
    ПОМЕСТИТЬ ОплатаПоПорядку
    ИЗ
    Оплата КАК Оплата
    ЛЕВОЕ СОЕДИНЕНИЕ Оплата КАК Оплата1
    ПО Оплата.Товар > Оплата1.Товар

    СГРУППИРОВАТЬ ПО
    Оплата.Товар,
    Оплата.Количество,
    Оплата.Дата
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    РеализацияПоПорядку.Ячейка КАК Оплата,
    ОплатаПоПорядку.Товар КАК Реализация,
    ВЫБОР
    КОГДА ОплатаПоПорядку.ПорядокРаспределенияПо < РеализацияПоПорядку.ПорядокРаспределенияПо
    ТОГДА ОплатаПоПорядку.ПорядокРаспределенияПо
    ИНАЧЕ РеализацияПоПорядку.ПорядокРаспределенияПо
    КОНЕЦ - ВЫБОР
    КОГДА ОплатаПоПорядку.ПорядокРаспределенияС > РеализацияПоПорядку.ПорядокРаспределенияС
    ТОГДА ОплатаПоПорядку.ПорядокРаспределенияС
    ИНАЧЕ РеализацияПоПорядку.ПорядокРаспределенияС
    КОНЕЦ + 1 КАК Количество,
    ОплатаПоПорядку.Дата
    ИЗ
    РеализацияПоПорядку КАК РеализацияПоПорядку
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ОплатаПоПорядку КАК ОплатаПоПорядку
    ПО РеализацияПоПорядку.ПорядокРаспределенияС <= ОплатаПоПорядку.ПорядокРаспределенияПо
    И РеализацияПоПорядку.ПорядокРаспределенияПо >= ОплатаПоПорядку.ПорядокРаспределенияС
```

И результат:
Оплата Реализация Количество Дата
Р1 О1 1 000 02.10.2016 0:00:00
Р2 О1 500 02.10.2016 0:00:00
Р2 О2 1 000 04.10.2016 0:00:00
Р2 О3 500 10.10.2016 0:00:00
Р3 О3 2 500 10.10.2016 0:00:00
---
спасибо большущее, вроде бы все получилось
сейчас еще проверю, но похоже то что надо.
Спасибо